



## 分析指南： 使用华盛顿小组详细功能问题集（WG-ES）Stata语法创建残疾标识符

### 导言

与华盛顿小组简易功能问题集 (WG-SS) 一样，对华盛顿小组详细功能问题集 (WG-ES) 的分析也可以根据严重性阈值或分界点的选择产生多个残疾标识符。但是，下面的Stata语法规定使用不同的WG-ES领域组合来计算残疾标识符，并使用建议的分界点以便进行国际比较（如下文所述）。

对于所述的每个残疾标识符，纳入程度为至少一个领域/问题编码为“非常困难”或“完全无法做到”，或者在焦虑、抑郁、疼痛和疲劳领域，为四分量表上最大程度的困难。

本文件所述的四种残疾标识符的定义均基于对功能领域的选择，包括：

**WG-SS:** 简易问题集：6个领域，6个问题

**WG-ES 1:** 详细问题集：11个领域，25个问题

**WG-ES 2:** 修订版详细问题集（WG-ES去除“疼痛”和“疲劳”这两个领域）：9个领域，20个问题

**WG-ES 3:** 简易问题集增强版（WG-SS加上“上半身机能”、“焦虑”和“抑郁”这三个领域）：9个领域，12个问题

**注：**在进行数据分析时，使用标准的权重和估算技术。

Stata语法基于下表所示变量标签。完整的WG-ES模块包含的问题多于此表中显示的问题。残疾状况是通过在没有使用辅助技术或其他协助的情况下，从事基本的、普遍的活动时出现的困难程度来确定的。例如，有几个关于活动度的问题提到在使用协助的情况下的行走困难程

**《华盛顿小组实施文件》**涵盖了华盛顿残疾统计小组(WG)开发的、用来在人口普查和调查中收集国际可比的残疾数据的多款工具。实施文件介绍了实施简易问题集、详细问题集、简易问题集-增强版、华盛顿小组/儿基会儿童功能模块(2-4岁和5-17岁)、华盛顿小组/国际劳工组织(ILO)劳动力调查残疾模块，以及其他华盛顿小组工具的最佳实践。主题包括翻译、问题规范、分析指南、用于分析的编程代码、分类工具的使用等等。

如需查找其他《华盛顿小组实施文件》和更多信息，请访问华盛顿小组网站：  
<http://www.washingtongroup-disability.com/>。

度。这些问题未包含在此处提供的分析计划中。但是，它们可以用于其他分析中，以更仔细地研究辅助技术（环境促进因素）对功能的影响。

只有以下这些问题/变量用于确定残疾标识符。**确保您使用的是相同的变量标签，或者修改Stata语法以反映在数据库中使用的变量标签。**

WG-SS作为美国国民健康访谈调查(NHIS)的一部分实施。用于编制这些指南的数据来自2013年的NHIS。

**NHIS用户注意：** NHIS数据文件和文档中的变量名称可能与本文档中使用的变量名称不同；例如，在本文档中称为SC-SS的自理领域变量在NHIS数据文件和文档中称为UB\_SS。

华盛顿小组详细问题集中的问题/领域	变量标签	回答模式
视力		
<b>1.即使戴着眼镜，您是否也难以看清楚？</b>	VIS_SS	1
沟通		
<b>2.您是否难以使用日常语言和他人沟通，如理解他人或被他人理解？</b>	COM_SS	1
听力		
<b>3.即使使用助听器，您是否也难以听清楚？</b>	HEAR_SS	1
<b>4.难以听清在安静房间内与他人谈话的内容？</b>	HEAR_3	1
<b>5.难以听清在更嘈杂房间内与他人谈话的内容？</b>	HEAR_4	1
认知		
<b>6.您是否难以记住东西或集中注意力？</b>	COG_SS	1
<b>7.难以记住东西、集中注意力或两种情况都有？</b>	COG_1	2
<b>8.难以记住东西的情况多久出现一次？</b>	COG_2	3
<b>9.难以记住的事情有多少？</b>	COG_3	4
自理/上半身		
<b>10.您是否难以自理生活，例如清洗全身或穿衣？</b>	SC_SS	1
<b>11.难以将一瓶2升的水从腰部举到齐眼高度？</b>	UB_1	1
<b>12.使用手和手指的困难程度</b>	UB_2	1

活动度		
13.您是否难以正常行走或上下阶梯?	MOB_SS	1
14.在未依靠协助或设备的情况下, 难以在平地行走100码?	MOB_4	1
15.在未依靠协助或设备的情况下, 难以在平地行走1/3英里?	MOB_5	1
16.在未依靠协助或设备的情况下, 难以向上或向下走12级阶梯?	MOB_6	1
情感 (焦虑)		
17.多久会出现感觉到担忧、紧张或焦虑的情况?	ANX_1	5
18.上一次感到担忧、紧张或焦虑时的感受程度?	ANX_3	6
情感 (抑郁)		
19.您多久会出现感觉到抑郁的情况?	DEP_1	5
20.上一次您感到抑郁时, 您感觉到有多抑郁?	DEP_3	6
疼痛		
21.过去3个月内疼痛发生的频率?	PAIN_2	7
22.上一次您感到疼痛时, 您感觉到有多疼痛?	PAIN_4	6
疲劳		
23.过去3个月内, 多久会出现感到非常疲劳或精疲力尽的情况?	TIRED_1	7
24.最近一次非常疲劳或精疲力尽的感觉持续了多长时间?	TIRED_2	8
25.最近一次感到非常疲劳或精疲力尽时, 疲劳的程度?	TIRED_3	6

注: 红色代表华盛顿小组简易问题集(WG-SS)。  
所有25个问题都包含在WG-ES 1中。

红色加蓝色的问题包含在WG-ES 2中。

红色加绿色的问题包括在WG-ES 3中。

## 回答模式

	模式1	模式2	模式3	模式4
1	没有困难	仅难以记住东西	有时	一些事
2	是，有点困难	仅难以集中注意力	经常	很多事
3	是，非常困难	记住东西和集中注意力均有困难	总是	几乎所有事
4	完全无法做到			
7	拒绝回答			
8	不确定			
9	不知道			

	模式5	模式6*	模式7	模式8
1	每天都有	有点	从未有过	当天的一段时间
2	每周都有	显著	有时候	当天的大多数时间
3	每月都有	介于有点和显著之间	大多数时候	全天
4	一年几次		每天	
5	从未有过			
7	拒绝回答			
8	不确定			
9	不知道			

\*在下面的语法中，请注意回答模式为6（ANX\_3、DEP\_3、PAIN\_4和TIRED\_3）的题目被重新编码，以便将“介于”的数值放置在“有点”和“显著”之间。

WG-SS嵌入到WG-ES中。WG-ES还补充了以下内容：

- 现有6个领域的额外问题，以及
- 额外领域（有几个领域具有多个问题）。

下面介绍的Stata语法包括一些WG-ES内容特有的元素。

首先，为包括多个问题的功能领域确定特定领域单一标识符是很重要的。例如，上半身机能包括两个问题，每个问题都引出了具体而独特的动作：将一瓶水从腰部举到齐眼高度（手臂/肩膀）的困难程度，以及使用手和手指的困难程度。对这两个问题进行分析和组合后，产生一个上半身指标，困难程度分为4级，从“1 - 困难程度低”到“4 - 困难程度高”——这与对单个

WG-SS问题的分类回答没有什么不同：没有困难、有点困难、非常困难、完全无法做到。与上半身领域一样，WG-ES领域“认知、焦虑、抑郁、疼痛和疲劳”有不同的回答模式，不易“转化”为通常的华盛顿小组回答模式。对于这些功能领域，编制了一个相似的4分量表回答模式，并标注为1级到4级，1表示困难程度最低，而4表示困难程度最高。

其次，对各个领域指标一起进行评估，以确定纳入总体残疾标识符的适当分界点——以便根据残疾状况估计发生率和对结果指标进行分类。

**注：**

对于所有变量，代码(7)“拒绝回答”，(8)“不确定”，以及(9)“不知道”，被重新编码为“缺失”。

**Stata WG Extended Set Syntax Annotated with Output Tables**

Actual Stata syntax is indented and are in **Bold** text.

NOTE: For data analysis, use your standard weighting and estimation techniques.

The syntax below produces **frequency distributions** on individual domain questions – **cross-tabulations** on multiple domain questions, and calculates INDICATOR variables for domains with multiple questions – for use in the determination of disability identifiers.

**VISION**

*Step 1. Generate frequency distribution for Vision domain.*

VIS\_SS is the WG-SS Vision question.

```
gen Vision=VIS_SS if inlist(VIS_SS, 1,2,3,4)
replace Vision=. if inlist(VIS_SS, 7,8,9)
tabulate Vision
```

**Vision: Degree of difficulty seeing**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13690	79.0	81.6	81.6
	Some difficulty	2708	15.6	16.2	97.8
	A lot of difficulty	333	1.9	2.0	99.8
	Cannot do at all	36	.2	.2	100.0
	Total	16767	96.8	100.0	
Missing		559	3.2		
Total		17326	100.0		

## COMMUNICATION

Step 2. Generate frequency distribution for Communication domain.

COM\_SS is the WG-SS Communication question.

```
gen Communication=COM_SS if inlist(COM_SS, 1,2,3,4)
replace Communication =. if inlist(COM_SS, 7,8,9)
tabulate Communication
```

### Communication: Degree of difficulty communicating using usual language

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15874	91.6	94.7	94.7
	Some difficulty	745	4.3	4.4	99.2
	A lot of difficulty	94	.5	.6	99.7
	Cannot do at all	43	.2	.3	100.0
	Total	16756	96.7	100.0	
Missing		570	3.3		
Total		17326	100.0		

## HEARING

Step 3. Generate frequency distributions and cross-tabulations for Hearing domain questions and determine Hearing Indicator

HEAR\_SS is the WG-SS Hearing question.

```
gen Hearing=HEAR_SS if inlist(HEAR_SS, 1,2,3,4)
replace Hearing=. if inlist(HEAR_SS, 7,8,9)
```

HEAR\_3 is *Difficulty hearing conversation with one person in quiet room.*

```
gen HEAR_3_R=HEAR_3 if inlist(HEAR_3, 1,2,3,4)
replace HEAR_3_R =. if inlist(HEAR_3, 7,8,9)
```

HEAR\_4 is *Difficulty hearing one person in noisier room.*

```
gen HEAR_4_R=HEAR_4 if inlist(HEAR_4, 1,2,3,4)
replace HEAR_4_R=. if inlist(HEAR_4, 7,8,9)
tabulate Hearing
```

### Hearing: Degree of difficulty hearing

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13680	79.0	81.6	81.6
	Some difficulty	2753	15.9	16.4	98.0
	A lot of difficulty	310	1.8	1.8	99.9
	Cannot do at all	23	.1	.1	100.0
	Total	16766	96.8	100.0	
Missing		560	3.2		
Total		17326	100.0		

**tabulate** HEAR\_3\_R

### HEAR\_3\_R: Difficulty hearing conversation with one person in quiet room

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15249	88.0	91.0	91.0
	Some difficulty	1316	7.6	7.9	98.9
	A lot of difficulty	162	.9	1.0	99.9
	Cannot do at all	10	.1	.1	100.0
	Total	16737	96.6	100.0	
Missing		589	3.4		
Total		17326	100.0		

**tabulate** HEAR\_4\_R

### HEAR\_4\_R: Difficulty hearing one person in noisier room

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	11697	67.5	70.0	70.0
	Some difficulty	4191	24.2	25.1	95.1
	A lot of difficulty	779	4.5	4.7	99.7
	Cannot do at all	48	.3	.3	100.0
	Total	16715	96.5	100.0	
Missing		611	3.6		
Total		17326	100.0		

Step 4. For Hearing questions, recode HEAR\_3\_R and HEAR\_4\_R to value 4 (cannot do at all) if Hear\_SS is 4 (Cannot do at all).

The syntax below recodes HEAR\_3\_R and HEAR\_4\_R to 4 (cannot do at all) if Hear\_SS is 4 (cannot do at all).

**gen** HEAR\_3\_X=HEAR\_3\_R

**replace** HEAR\_3\_X=4 **if** Hearing==4 & HEAR\_3\_R==.

**gen** HEAR\_4\_X=HEAR\_4\_R

**replace** HEAR\_4\_X=4 **if** Hearing==4 & HEAR\_4\_R==.

**tabulate** HEAR\_3\_X

**HEAR\_3\_X: Difficulty hearing conversation with one person in quiet room**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15249	88.0	91.0	91.0
	Some difficulty	1316	7.6	7.9	98.8
	A lot of difficulty	162	.9	1.0	99.8
	Cannot do at all	33	.2	.2	100.0
	Total	16760	96.7	100.0	
Missing		566	3.3		
Total		17326	100.0		

**tabulate** HEAR\_4\_X

**HEAR\_4\_X: Difficulty hearing one person in noisier room**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	11697	67.5	69.9	69.9
	Some difficulty	4191	24.2	25.0	94.9
	A lot of difficulty	779	4.5	4.7	99.6
	Cannot do at all	71	.4	.4	100.0
	Total	16738	96.6	100.0	
Missing		588	3.4		
Total		17326	100.0		

Step 5. Generate a cross-tabulation of the two Hearing Extended Set questions: HEAR\_3\_X and HEAR\_4\_X.

The syntax below produces a cross-tabulation of the two Extended Set questions: HEAR\_3\_X: *Difficulty hearing in a quiet room* and HEAR\_4\_X: *Difficulty hearing in a noisier room* to determine a single HEARING INDICATOR.

**tabulate** HEAR\_4\_X HEAR\_3\_X



**HEAR\_3\_X : Difficulty hearing conversation with one person  
in quiet room**

<b>HEAR_4_X (Difficulty hearing in a Nosier room)</b>		<b>HEAR_3_X : Difficulty hearing conversation with one person in quiet room</b>				<b>Total</b>
		<b>No difficulty</b>	<b>Some difficulty</b>	<b>A lot of difficulty</b>	<b>Cannot do at all</b>	
No difficulty	<b>11603</b>	<b>94</b>	<b>0</b>	<b>0</b>	11697	
Some difficulty	<b>3373</b>	<b>809</b>	<b>8</b>	<b>0</b>	4190	
A lot of difficulty	<b>253</b>	<b>388</b>	<b>138</b>	<b>0</b>	779	
Cannot do at all	<b>8</b>	<b>24</b>	<b>16</b>	<b>23</b>	71	
<b>Total</b>	15237	1315	162	23	16737	

Step 6. Create a HEARING INDICATOR (H\_INDICATOR) based on the two additional hearing questions HEAR\_3\_X and HEAR\_4\_X.

The syntax below creates a HEARING INDICATOR (H\_INDICATOR) based on the cross-tabulation of the two additional hearing questions HEAR\_3\_X and HEAR\_4\_X.

```

gen H_INDICATOR=.
replace H_INDICATOR=1 if (HEAR_3_X==1 & HEAR_4_X==1) | ///
(HEAR_3_X==1 & HEAR_4_X==2)
replace H_INDICATOR=2 if (HEAR_3_X==2 & (HEAR_4_X==1 | ///
HEAR_4_X==2)) | (HEAR_3_X==1 & HEAR_4_X==3)
replace H_INDICATOR=3 if (HEAR_3_X==3 & (HEAR_4_X==1 | ///
HEAR_4_X==2) | (HEAR_3_X==2 & HEAR_4_X==3) | ///
(HEAR_3_X==1 & HEAR_4_X==4))
replace H_INDICATOR=4 if ((HEAR_3_X==3 & HEAR_4_X==3) | ///
HEAR_3_X==4 | (HEAR_4_X==4 & (HEAR_3_X==2 | HEAR_3_X==3)))
tabulate H_INDICATOR

```

		<b>H_INDICATOR</b>			
		<b>Frequency</b>	<b>Percent</b>	<b>Valid Percent</b>	<b>Cumulative Percent</b>
Valid	<b>1.00</b>	14976	86.4	89.4	89.4
	<b>2.00</b>	1156	6.7	6.9	96.3
	<b>3.00</b>	404	2.3	2.4	98.7
	<b>4.00</b>	211	1.2	1.3	100.0
	<b>Total</b>	16747	96.7	100.0	
Missing		579	3.3		
<b>Total</b>		17326	100.0		

**COGNITION: Degree of difficulty remembering or concentrating**

Step 7. Generate frequency distributions and cross-tabulations for Cognition domain questions and determine a Cognition Indicator.

```
gen Cognition=COG_SS if inlist(COG_SS, 1,2,3,4)
replace Cognition=. if inlist(COG_SS, 7,8,9)
tabulate Cognition
```

**Cognition: Degree of difficulty remembering or concentrating**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13719	79.2	81.9	81.9
	Some difficulty	2632	15.2	15.7	97.6
	A lot of difficulty	382	2.2	2.3	99.9
	Cannot do at all	20	.1	.1	100.0
	Total	16753	96.7	100.0	
Missing		573	3.3		
Total		17326	100.0		

If response to COG-SS is *some difficulty*, *a lot of difficulty* or *cannot do at all*, then the respondent is asked COG\_1: whether they have difficulty remembering, concentrating or both.

```
gen COG_1_R=COG_1 if inlist(COG_1, 1,2,3)
replace COG_1_R=. if inlist(COG_1, 7,8,9)
tabulate COG_1_R
```

**COG\_1\_R: Difficulty remembering, concentrating, or both?**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Difficulty remembering only	983	5.7	32.4	32.4
	Difficulty concentrating only	388	2.2	12.8	45.2
	Difficulty with both remembering and concentrating	1659	9.6	54.8	100.0
	Total	3030	17.5	100.0	
Missing		14296	82.5		
Total		17326	100.0		

Step 8. Account for those who did not answer COG\_1 (COG\_SS is 1 – no difficulty and they were skipped) by recoding COG\_1 to 0 (No difficulty).

If response to COG-SS is 1: *no difficulty*, then the variable COG\_1\_R is recoded into COG\_1A, and the value assigned is 0: *no difficulty*.

```
gen COG_1A=COG_1_R
replace COG_1A=0 if COG_SS==1
tabulate COG_1A
```

**COG\_1A: Difficulty remembering, concentrating, or both? (including no difficulty)**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13719	79.2	81.9	81.9
	Difficulty remembering only	983	5.7	5.9	87.8
	Difficulty concentrating only	388	2.2	2.3	90.1
	Difficulty with both remembering and concentrating	1659	9.6	9.9	100.0
	Total	16749	96.7	100.0	
Missing		577	3.3		
Total		17326	100.0		

Step 9. Generate frequency distribution for remaining cognition questions.

Frequency distribution of the Cognition extended REMEMBERING questions: COG\_2 *How often have difficulty remembering*, and COG\_3 *Amount of things you have difficulty remembering*.

```
gen COG_2_R=COG_2 if inlist(COG_2, 1,2,3)
replace COG_2_R=. if inlist(COG_2, 7,8,9)

gen COG_3_R=COG_3 if inlist(COG_3, 1,2,3)
replace COG_3_R=. if inlist(COG_3, 7,8,9)
tabulate COG_2_R
```

**COG\_2\_R: How often have difficulty remembering?**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Sometimes	1916	11.1	72.4	72.4
	Often	513	3.0	19.4	91.8
	All of the time	216	1.2	8.2	100.0
	Total	2645	15.3	100.0	
Missing		14681	84.7		
Total		17326	100.0		

**tabulate** COG\_3\_R

**COG\_3\_R: Amount of things you have difficulty remembering?**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	A few things	2119	12.2	80.3	80.3
	A lot of things	386	2.2	14.6	94.9
	Almost everything	134	.8	5.1	100.0
	Total	2639	15.2	100.0	
Missing		14687	84.8		
Total		17326	100.0		

Step 10. Generate cross-tabulation of the two Cognition extended set questions COG\_2R by COG\_3\_R.

The syntax below produces a cross-tabulation of the two Extended Set REMEMBERING questions: COG\_2\_R: *How often you have difficulty remembering* and COG\_3\_R: *The amount of things you have difficulty remembering* to determine a single REMEMBERING INDICATOR.

**tabulate** COG\_2\_R COG\_3\_R

		COG_3_R: Amount of things you have difficulty remembering?				
		A few things	A lot of things	Almost everything	Total	
COG_2_R: How often do you have difficulty remembering?	How often have difficulty remembering?	Sometimes	<b>1788</b>	<b>105</b>	<b>20</b>	1913
		Often	<b>279</b>	<b>197</b>	<b>34</b>	510
		All of the time	<b>51</b>	<b>84</b>	<b>80</b>	215
Total			2118	386	134	2638

Step 11. Create a Remembering Indicator based on distribution of COG\_2\_R and COG\_3\_R.

The syntax below creates a REMEMBERING INDICATOR (R\_INDICATOR) based on the two additional remembering questions (COG\_2\_R and COG\_3\_R).

If Cognition is 1: no difficulty, then the Remembering Indicator is coded as 1: the lowest level of difficulty.

```

gen R_INDICATOR=0
replace R_INDICATOR=1 if Cognition==1
replace R_INDICATOR=2 if ((COG_2_R==1 & COG_3_R==1) | ///
    (COG_3_R==1 & COG_2_R==2) | (COG_3_R==2 & COG_2_R==1))
replace R_INDICATOR=3 if (COG_3_R==2 & COG_2_R==2)
replace R_INDICATOR=4 if (COG_3_R==3 | COG_2_R==3)
  
```

Step 12. If COG\_1A is coded as 2 (concentrating only), then the Remembering Indicator is coded as 5.

These 388 individuals are respondents who were not included in the Remembering Indicator since they had only difficulty concentrating.

**replace** R\_INDICATOR=5 **if** (COG\_1A==2)

Step 13. Generate frequency distribution of the Remembering Indicator.

**tabulate** R\_INDICATOR

		<b>R_INDICATOR</b>			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	.00	580	3.3	3.3	3.3
	<b>1.00</b>	13719	79.2	79.2	82.5
	<b>2.00</b>	2172	12.5	12.5	95.1
	<b>3.00</b>	197	1.1	1.1	96.2
	<b>4.00</b>	270	1.6	1.6	97.8
	5.00	388	2.2	2.2	100.0
Total		17326	100.0	100.0	

Step 14. Supplement Remembering Indicator with information on difficulty concentrating.

The syntax below adds information on whether cognitive difficulties are compounded by difficulty concentrating in addition to difficulty remembering.

Create a COGNITION INDICATOR (COG\_INDICATOR) based on R\_INDICATOR (above) and the cognition question (COG\_1\_R).

The 388 individuals with ‘concentrating only’ were allocated as follows:

1. 357 with a little difficulty on Cognition question were classified as **2**
2. 30 with a lot of difficulty on Cognition question were classified as **3**
3. 1 with cannot do on Cognition question was classified as **4**

Those with both remembering and concentrating difficulty were upgraded 36 individuals from **2** to **3**, and 125 individuals from **3** to **4**.

**gen** COG\_INDICATOR=R\_INDICATOR

**replace** COG\_INDICATOR=2 **if** (R\_INDICATOR==5 & Cognition==2)

**replace** COG\_INDICATOR=3 **if** (R\_INDICATOR==5 & Cognition==3)

**replace** COG\_INDICATOR=4 **if** (R\_INDICATOR==5 & Cognition==4)

**replace** COG\_INDICATOR=3 **if** (R\_INDICATOR==2 & COG\_1\_R==3 & Cognition==3)

**replace** COG\_INDICATOR=4 **if** (R\_INDICATOR==3 & COG\_1\_R==3 & Cognition==3)

Step 15. Generate frequency distribution of the Cognition Indicator.

**tabulate** COG\_INDICATOR

		COG_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	.00	580	3.3	3.3	3.3
	1.00	13719	79.2	79.2	82.5
	2.00	2449	14.1	14.1	96.7
	3.00	226	1.3	1.3	98.0
	4.00	352	2.0	2.0	100.0
Total		17326	100.0	100.0	

## UPPER BODY

Step 16. Generate frequency distributions and cross-tabulations for Self-care and Upper body domain questions and determine the Upper Body Indicator.

SC\_SS is the WG-SS Self-care question.

UB\_1 is *Difficulty raising 2 liter bottle of water from waist to eye level.*

UB\_2 is *Difficulty using hands and fingers*

**gen** SELF\_CARE=SC\_SS **if inlist**(SC\_SS, 1,2,3,4)  
**replace** SELF\_CARE=. **if inlist**(SC\_SS, 7,8,9)

**gen** UB\_1\_R=UB\_1 **if inlist**(UB\_1, 1,2,3,4)  
**replace** UB\_1\_R=. **if inlist**(UB\_1, 7,8,9)

**gen** UB\_2\_R=UB\_2 **if inlist**(UB\_2, 1,2,3,4)  
**replace** UB\_2\_R=. **if inlist**(UB\_2, 7,8,9)

First, calculate frequency distributions on the short set and two extended set questions.

**tabulate** SELF\_CARE

		SELF_CARE: Degree of difficulty with self-care			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	16029	92.5	95.7	95.7
	Some difficulty	544	3.1	3.2	98.9
	A lot of difficulty	114	.7	.7	99.6
	Cannot do at all	68	.4	.4	100.0
	Total	16755	96.7	100.0	
Missing		571	3.3		
Total		17326	100.0		

`tabulate UB_1_R`

**UB\_1\_R: Diff raising 2 liter bottle of water from waist to eye level**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15677	90.5	93.6	93.6
	Some difficulty	743	4.3	4.4	98.0
	A lot of difficulty	167	1.0	1.0	99.0
	Cannot do at all	166	1.0	1.0	100.0
	Total	16753	96.7	100.0	
Missing		573	3.3		
Total		17326	100.0		

`tabulate UB_2_R`

**UB\_2\_R: Degree of difficulty using hands and fingers**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	15199	87.7	90.7	90.7
	Some difficulty	1229	7.1	7.3	98.1
	A lot of difficulty	255	1.5	1.5	99.6
	Cannot do at all	70	.4	.4	100.0
	Total	16753	96.7	100.0	
Missing		573	3.3		
Total		17326	100.0		

Step 17. Generate a cross-tabulation of the two Upper body Extended Set questions: UB\_2\_R and UB\_1\_R.

The syntax below produces a cross-tabulation of the two Extended Set questions: UB\_1\_R: *Difficulty raising a 2 liter bottle of water from waste to eye level* UB\_2\_R: *Difficulty using hands and fingers* to determine a single UPPER BODY INDICATOR (UB\_INDICATOR).

`tabulate UB_2_R UB_1_R`

		UB_1_R: Diff raising 2 liter bottle of water from waist to eye level				Total
		No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	
UB_2_R: Difficulty using hands and fingers	No difficulty	14786	309	58	44	15197
	Some difficulty	782	355	51	40	1228
	A lot of difficulty	98	73	51	33	255
	Cannot do at all	9	5	7	49	70
Total		15675	742	167	166	16750

Step 18. Create an UPPER BODY INDICATOR (UB\_INDICATOR) based on the two additional upper body questions UB\_2\_R and UB\_3\_R.

Syntax below creates UB\_INDICATOR based on the distribution in the cross-tabulation above.

```
gen UB_INDICATOR=4 if (UB_1_R==4 | UB_2_R==4)
replace UB_INDICATOR=3 if UB_INDICATOR ~=4 & (UB_1_R==3 | UB_2_R==3)
replace UB_INDICATOR=2 if UB_INDICATOR ~=4 & UB_INDICATOR~=3 & ///
(UB_1_R==2 | UB_2_R==2)
replace UB_INDICATOR=1 if UB_INDICATOR~=4 & UB_INDICATOR~=3 & ///
UB_INDICATOR~=2 & (UB_1_R==1 | UB_2_R==1)
tabulate UB_INDICATOR
```

		UB_INDICATOR			
		Frequen cy	Percent	Valid Percent	Cumulative Percent
Valid	1.00	14790	85.4	88.3	88.3
	2.00	1448	8.4	8.6	96.9
	3.00	331	1.9	2.0	98.9
	4.00	187	1.1	1.1	100.0
	Total	16756	96.7	100.0	
Missing		570	3.3		
Total		17326	100.0		

## MOBILITY

Step 19. Generate frequency distributions and cross-tabulations for Mobility domain questions and determine Mobility Indicator.

MOB\_SS is the WG-SS Mobility question.

MOB\_4 is Difficulty walking 100 yards on level ground without aid or equipment.

MOB\_5 is Difficulty walking 1/3rd mile on level ground without aid or equipment.

```
gen Mobility=MOB_SS if inlist(MOB_SS, 1,2,3,4)
replace Mobility=. if inlist(MOB_SS, 7,8,9)
```

```
gen MOB_4_R=MOB_4 if inlist(MOB_4, 1,2,3,4)
replace MOB_4_R=. if inlist(MOB_4, 7,8,9)
```

```
gen MOB_5_R=MOB_5 if inlist(MOB_5, 1,2,3,4)
replace MOB_5_R=. if inlist(MOB_5, 7,8,9)
```

First, calculate frequency distributions on the short set and two extended set WALKING questions.



**tabulate** Mobility

**Mobility: Degree of difficulty walking or climbing steps**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13424	77.5	80.1	80.1
	Some difficulty	2165	12.5	12.9	93.0
	A lot of difficulty	792	4.6	4.7	97.7
	Cannot do at all	380	2.2	2.3	100.0
	Total	16761	96.7	100.0	
Missing		565	3.3		
Total		17326	100.0		

**tabulate** MOB\_4\_R

**MOB\_4\_R: Diff walking 100 yards on level ground w/o aid or equipment**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13892	80.2	84.8	84.8
	Some difficulty	1369	7.9	8.4	93.2
	A lot of difficulty	491	2.8	3.0	96.2
	Cannot do at all	<b>623</b>	3.6	3.8	100.0
	Total	16375	94.5	100.0	
Missing		951	5.5		
Total		17326	100.0		

**tabulate** MOB\_5\_R

**MOB\_5\_R: Diff walking 1/3rd mile on level ground w/o aid or equipment**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	No difficulty	13025	75.2	82.8	82.8
	Some difficulty	1650	9.5	10.5	93.3
	A lot of difficulty	708	4.1	4.5	97.8
	Cannot do at all	347	2.0	2.2	100.0
	Total	15730	90.8	100.0	
Missing		1596	9.2		
Total		17326	100.0		

Step 20. Generate a cross-tabulation of the walking distance questions: MOB\_4\_R and MOB\_5\_R.

The syntax below produces a cross-tabulation of the two Extended Set WALKING questions: MOB\_4\_R: *Difficulty walking 100 yards without equipment* and MOB\_5\_R: *Difficulty walking 1/3 mile without equipment* to determine a single WALKING INDICATOR.

NOTE: **623** individuals who responded cannot do at all to MOB\_4\_R were not asked MOB\_5\_R and they do not appear in the table below. They are, however, accounted for in the WALKING indicator calculation.

**tabulate** MOB\_4\_R MOB\_5\_R

MOB_4_R: Diff walking 100 yards on level ground w/o aid or equipment	MOB_5_R: Diff walking 1/3rd mile on level ground w/o aid or equipment				Total
	No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	
No difficulty	<b>12950</b>	<b>819</b>	<b>63</b>	<b>39</b>	13871
Some difficulty	<b>72</b>	<b>810</b>	<b>343</b>	<b>142</b>	1367
A lot of difficulty	<b>3</b>	<b>21</b>	<b>301</b>	<b>166</b>	491
Cannot do at all ( <b>623</b> )	<b>0</b>	<b>0</b>	<b>0</b>	<b>0</b>	0
Total	13025	1650	707	347	15729

Step 21. Create a WALKING INDICATOR (WALK\_INDICATOR) based on the two additional walking questions MOB\_4\_R and MOB\_5\_R.

Syntax below creates WALKING\_INDICATOR based on the distribution in the cross-tabulation above.

```

gen WALK_INDICATOR=0
replace WALK_INDICATOR=1 if (MOB_4_R==1 & (MOB_5_R==1 | MOB_5_R==2))
replace WALK_INDICATOR=2 if (MOB_4_R==1 & MOB_5_R==3) | (MOB_4_R==2 & ///
(MOB_5_R==1 | MOB_5_R== 2 | MOB_5_R==3))
replace WALK_INDICATOR=3 if (MOB_4_R==1 & MOB_5_R==4) | (MOB_4_R==3 & ///
(MOB_5_R==1 | MOB_5_R==2 | MOB_5_R==3))
replace WALK_INDICATOR=4 if (MOB_4_R==2 & MOB_5_R==4) | (MOB_4_R==3 & ///
MOB_5_R==4)

```

Syntax below includes the **623** who responded cannot do at all to MOB\_4\_R into the WALKING INDICATOR.

```

replace WALK_INDICATOR=4 if (WALK_INDICATOR==0 & MOB_4_R==4)
replace WALK_INDICATOR=. if WALK_INDICATOR==0
tabulate WALK_INDICATOR

```

### WALK\_INDICATOR

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	<b>1.00</b>	13769	79.5	84.2	84.2
	<b>2.00</b>	1288	7.4	7.9	92.1
	<b>3.00</b>	364	2.1	2.2	94.3
	<b>4.00</b>	931	5.4	5.7	100.0
	Total	16352	94.4	100.0	
Missing		974	5.6		
Total		17326	100.0		

Step 22. Supplement Walking Indicator with information on difficulty Climbing steps (MOB\_6).

Syntax below adds information from MOB\_6 on *difficulty climbing up or down 12 steps* to create a combined Mobility Indicator (MOB\_INDICATOR).

```
gen MOB_6_R=MOB_6 if inlist(MOB_6, 1,2,3,4)
replace MOB_6_R=. if inlist(MOB_6, 7,8,9)
tabulate WALK_INDICATOR MOB_6_R
```

		MOB_6_R: Difficulty climbing up or down 12 steps				
WALK_INDICATOR		No difficulty	Some difficulty	A lot of difficulty	Cannot do at all	Total
walk_INDICATOR2	<b>1.00</b>	13048	645	<b>55</b>	<b>17</b>	13765
	<b>2.00</b>	370	767	<b>135</b>	<b>16</b>	1288
	<b>3.00</b>	43	106	200	14	363
	<b>4.00</b>	79	242	226	384	931
Total		13540	1760	616	431	16347

According to the table above, the syntax below reclassifies:

1. 55 individuals with level **1** on the WALKING INDICATOR as level **2** on the MOBILITY INDICATOR
2. 17 individuals with level **1** on the WALKING INDICATOR and 135 individuals with level **2** on the WALKING INDICATOR as level **3** on the MOBILITY INDICATOR, and
3. 16 individuals with level **2** on the WALKING INDICATOR as level **4** on the MOBILITY INDICATOR.

Step 23. Create a mobility indication (MOB\_INDICATOR) with information garnered from cross-tabulation above.

```
gen MOB_INDICATOR= WALK_INDICATOR
replace MOB_INDICATOR=2 if (WALK_INDICATOR==1 & MOB_6_R==3)
```

```

replace MOB_INDICATOR=3 if (WALK_INDICATOR==2 & MOB_6_R==3) | ///
(WALK_INDICATOR==1 & MOB_6_R==4)
replace MOB_INDICATOR=4 if (WALK_INDICATOR==2 & MOB_6_R==4)
tabulate MOB_INDICATOR

```

		MOB_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	13697	79.1	83.8	83.8
	2.00	1192	6.9	7.3	91.1
	3.00	516	3.0	3.2	94.2
	4.00	947	5.5	5.8	100.0
	Total	16352	94.4	100.0	
Missing		974	5.6		
Total		17326	100.0		

## ANXIETY

Step 24. Generate frequency distribution on ANX\_1.

First, calculate frequency distributions on ANX\_1: *How often do you feel worried, nervous or anxious?*

```

gen ANX_1_R=ANX_1 if inlist(ANX_1, 1,2,3,4,5)
replace ANX_1_R=. if inlist(ANX_1, 7,8,9)
tabulate ANX_1_R

```

### ANX\_1\_R: How often feel worried, nervous, or anxious?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Daily	1632	9.4	9.8	9.8
	Weekly	1872	10.8	11.2	21.0
	Monthly	1558	9.0	9.3	30.4
	A few times a year	4898	28.3	29.4	59.7
	Never	6714	38.8	40.3	100.0
	Total	16674	96.2	100.0	
Missing		652	3.8		
Total		17326	100.0		

Step 25. The syntax below recodes ANX\_3R into ANX\_3Y

- 1) to create a NOT ASKED category based on those who responded NEVER to ANX\_1\_R and
- 2) to place "SOMEWHERE BETWEEN" numerically in-between "A LITTLE" and "A LOT".

```

gen ANX_3Y=1 if ANX_3R==1

```

replace ANX\_3Y=3 if ANX\_3R==2  
 replace ANX\_3Y=2 if ANX\_3R==3  
 replace ANX\_3Y=. if inlist(ANX\_3R, 7, 8, 9)

Recode ANX\_3Y to 0 (not asked) if ANX\_1 is 5 (Never)

replace ANX\_3Y=0 if ANX\_1==5  
 tabulate ANX\_3Y

**ANX\_3Y: Level of feelings last time felt worried/nervous/anxious**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	6714	38.8	40.3	40.3
	A little	5700	32.9	34.2	74.5
	In between a little and a lot	3076	17.8	18.5	92.9
	A lot	1176	6.8	7.1	100.0
	Total	16666	96.2	100.0	
Missing		660	3.8		
Total		17326	100.0		

Step 26. Generate a cross-tabulation of the anxiety Extended Set questions: ANX\_1\_R and ANX\_3Y.

The syntax below produces a cross-tabulation of ANX\_1\_R: *How often you felt worried, nervous or anxious* (a measure of frequency) and ANX\_3Y: *The level of those feeling the last time you felt worried, nervous or anxious* (a measure of intensity) – used to determine a single ANXIETY INDICATOR (ANX\_INDICATOR).

tabulate ANX\_3Y ANX\_1\_R

ANX_3Y: Level of feelings last time felt worried, nervous or anxious	ANX_1_R: How often feel worried, nervous or anxious?					
	Daily	Weekly	Monthly	A Few Times A Year	Never	Total
Not asked	0	0	0	0	6714	6714
A little	489	887	897	3417	0	5690
In between a little and a lot	589	725	535	1221	0	3070
A lot	548	256	123	248	0	1175
Total	1626	1868	1555	4886	6714	16649

Step 27. Create an ANXIETY INDICATOR (ANX\_INDICATOR) based on the two anxiety questions ANX\_1\_R and ANX\_3Y.

Syntax below creates ANX\_INDICATOR based on the distribution in the cross-tabulation above.

```

gen ANX_INDICATOR=1 if (ANX_3Y <= 4 & (ANX_1_R==4 | ANX_1_R==5))
replace ANX_INDICATOR=2 if ((ANX_1_R==3) | (ANX_1_R < 3 & ANX_3Y==1) | ///
(ANX_1_R==2 & ANX_3Y==2))
replace ANX_INDICATOR=3 if ((ANX_1_R==1 & ANX_3Y==2) | ///
(ANX_1_R==2 & ANX_3Y==3))
replace ANX_INDICATOR=4 if (ANX_1_R==1 & ANX_3Y==3)
replace ANX_INDICATOR=. if (missing(ANX_1_R) | missing(ANX_3Y))
tabulate ANX_INDICATOR

```

### ANX\_INDICATOR

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	11600	67.0	69.7	69.7
	2.00	3656	21.1	22.0	91.6
	3.00	845	4.9	5.1	96.7
	4.00	548	3.2	3.3	100.0
	Total	16649	96.1	100.0	
Missing		677	3.9		
Total		17326	100.0		

## DEPRESSION

Step 28. Generate frequency distribution on DEP\_1.

First, calculate frequency distributions on DEP\_1: *How often do you feel depressed?*

```

gen DEP_1_R=DEP_1 if inlist(DEP_1, 1,2,3,4,5)
replace DEP_1_R=. if inlist(DEP_1, 7,8,9)
tabulate DEP_1_R

```

### DEP\_1\_R: How often do you feel depressed?

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Daily	756	4.4	4.5	4.5
	Weekly	926	5.3	5.6	10.1
	Monthly	1038	6.0	6.2	16.3
	A few times a year	4012	23.2	24.1	40.4
	Never	9929	57.3	59.6	100.0
	Total	16661	96.2	100.0	
Missing		665	3.8		
Total		17326	100.0		

Step 29. The syntax below recodes DEP\_3R into DEP\_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”. It also creates the category NOT ASKED, if DEP\_1 is NEVER (1)

```

gen DEP_3Y=1 if DEP_3R==1
replace DEP_3Y=3 if DEP_3R==2
replace DEP_3Y=2 if DEP_3R==3
replace DEP_3Y=. if inlist(DEP_3R, 7,8,9)

replace DEP_3Y=0 if DEP_1==5
tabulate DEP_3Y

```

**DEP\_3Y: Level of feelings last time felt depressed**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	9929	57.3	59.6	59.6
	A little	3775	21.8	22.7	82.3
	In between a little and a lot	2016	11.6	12.1	94.4
	A lot	935	5.4	5.6	100.0
	Total	16655	96.1	100.0	
Missing		671	3.9		
Total		17326	100.0		

Step 30. Generate a cross-tabulation of the depression Extended Set questions: DEP\_1\_R and DEP\_3Y.

The syntax below produces a cross-tabulation of DEP\_1\_R: *How often do you feel depressed* (a measure of frequency) and DEP\_3Y: *The level of those feeling the last time you felt depressed* (a measure of intensity) – used to determine a single DEPRESSION INDICATOR (DEP\_INDICATOR).

```

tabulate DEP_3Y DEP_1_R

```

DEP_3Y: Level of feelings last time felt depressed	DEP_1_R: How often do you feel depressed?					Total
	Daily	Weekly	Monthly	A Few Times A Year	Never	
Not asked	0	0	0	0	9929	9929
A little	161	346	548	2708	0	3763
In between a little and a lot	209	384	378	1042	0	2013
A lot	381	191	112	248	0	932
Total	751	921	1038	3998	9929	16637

Step 31. Create a DEPRESSION INDICATOR (DEP\_INDICATOR) based on the two depression questions DEP\_1\_R and DEP\_3Y.

Syntax below creates DEP\_INDICATOR based on the distribution in the cross-tabulation above.

```

gen DEP_INDICATOR=. if (missing(DEP_1_R) | missing(DEP_3Y))
replace DEP_INDICATOR=1 if (DEP_3Y <= 4 & (DEP_1_R==4 | DEP_1_R==5))
replace DEP_INDICATOR=2 if ((DEP_1_R==3) | (DEP_1_R < 3 & DEP_3Y==1) | ///
(DEP_1==2 & DEP_3Y==2))
replace DEP_INDICATOR=3 if ((DEP_1_R==1 & DEP_3Y==2) | ///
(DEP_1_R==2 & DEP_3Y==3))
replace DEP_INDICATOR=4 if (DEP_1_R==1 & DEP_3Y==3)
tabulate DEP_INDICATOR

```

**DEP\_INDICATOR**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	13927	80.4	83.7	83.7
	2.00	1929	11.1	11.6	95.3
	3.00	400	2.3	2.4	97.7
	4.00	381	2.2	2.3	100.0
	Total	16637	96.0	100.0	
Missing		689	4.0		
Total		17326	100.0		

**PAIN**

Step 32. Generate frequency distribution on PAIN\_2.

First, calculate frequency distributions on PAIN\_2: *Frequency of pain in the past 3 months.*

```

gen PAIN_2_R=PAIN_2 if inlist(PAIN_2, 1,2,3,4)
replace PAIN_2_R=. if inlist(PAIN_2, 7,8,9)
tabulate PAIN_2_R

```

**PAIN\_2\_R: Frequency of pain in past 3 months**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Never	6636	38.3	39.8	39.8
	Some days	6556	37.8	39.3	79.2
	Most days	1227	7.1	7.4	86.5
	Every day	2245	13.0	13.5	100.0
	Total	16664	96.2	100.0	
Missing		662	3.8		
Total		17326	100.0		



Step 33. The syntax below recodes PAIN\_4 into PAIN\_4Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”. It also creates the category NOT ASKED, if PAIN\_2 is NEVER (1).

```
gen PAIN_4Y=1 if PAIN_4==1
replace PAIN_4Y=3 if PAIN_4==2
replace PAIN_4Y=2 if PAIN_4==3
replace PAIN_4Y=. if inlist(PAIN_4, 7,8,9)

replace PAIN_4Y=0 if PAIN_2==1
tabulate PAIN_4Y
```

**PAIN\_4Y: How much pain you had last time you had pain?**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	6636	38.3	39.8	39.8
	A little	4865	28.1	29.2	69.0
	In between a little and a lot	3296	19.0	19.8	88.8
	A lot	1869	10.8	11.2	100.0
	Total	16666	96.2	100.0	
Missing		660	3.8		
Total		17326	100.0		

Step 34. Generate a cross-tabulation of the PAIN Extended Set questions: PAIN\_2\_R and PAIN\_4Y.

The syntax below produces a cross-tabulation of PAIN\_2\_R: *Frequency of pain in the past 3 months* and PAIN\_4Y: *How much pain you has the last time you had pain* (a measure of intensity) – used to determine a single PAIN INDICATOR (P\_INDICATOR).

```
tabulate PAIN_4Y PAIN_2_R
```

PAIN_4Y: How much pain you had last time you had pain		PAIN_2_R: Frequency of pain in past 3 months				
		Never	Some Days	Most days	Every day	Total
Not asked		6636	0	0	0	6636
A little		0	4136	323	401	4860
In between a little and a lot		0	1772	624	896	3296
A lot		0	645	278	944	1867
Total		6636	6553	1225	2241	16655

Step 35. Create a PAIN INDICATOR (P\_INDICATOR) based on the two PAIN questions PAIN\_2\_R and PAIN\_4Y.

Syntax below creates P\_INDICATOR based on the distribution in the cross-tabulation above.

```

gen P_INDICATOR=1 if (PAIN_2_R==1) | (PAIN_4Y==1 & ///
(PAIN_2_R==2 | PAIN_2_R==3))
replace P_INDICATOR=2 if ((PAIN_2_R==2 & (PAIN_4Y==2 | PAIN_4Y==3)) | ///
(PAIN_2_R==3 & PAIN_4Y==2) | (PAIN_2_R==4 & PAIN_4Y==1))
replace P_INDICATOR=3 if (PAIN_2_R==3 & PAIN_4Y==3) | ///
(PAIN_2_R==4 & PAIN_4Y==2)
replace P_INDICATOR=4 if (PAIN_2_R==4 & PAIN_4Y==3)
tabulate P_INDICATOR

```

		P_INDICATOR			
		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	1.00	11095	64.0	66.6	66.6
	2.00	3442	19.9	20.7	87.3
	3.00	1174	6.8	7.0	94.3
	4.00	944	5.4	5.7	100.0
	Total	16655	96.1	100.0	
Missing		671	3.9		
Total		17326	100.0		

## FATIGUE (Tired)

Step 36. Generate frequency distribution on FATIGUE Extended Set questions Tired\_1, Tired\_2 and Tired\_3.

First, calculate frequency distributions on TIRED\_1: *How often you felt tired in the past 3 months.*

```

gen TIRED_1_R=TIRED_1 if inlist(TIRED_1, 1,2,3,4)
replace TIRED_1_R=. if inlist(TIRED_1, 7,8,9)
tabulate TIRED_1_R

```

### TIRED\_1\_R: How often felt very tired or exhausted in past 3 months

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Never	5619	32.4	33.7	33.7
	Some days	8391	48.4	50.4	84.1
	Most days	1632	9.4	9.8	93.9
	Every day	1019	5.9	6.1	100.0
	Total	16661	96.2	100.0	
Missing		665	3.8		
Total		17326	100.0		

Step 37. Recode Tired\_2 to 0 (not asked) if Tired\_1 is 1 (Never).

If response to TIRED\_1 is 1: Never, then TIRED\_2 (*How long most recent tired or exhausted feelings lasted*) is not asked. This variable is recoded so these individuals are included in the syntax below.

<http://www.washingtongroup-disability.com/>

```

gen TIRED_2_R=.
replace TIRED_2_R=TIRED_2 if inlist(TIRED_2, 1,2,3)
replace TIRED_2_R=. if inlist(TIRED_2, 7,8,9)
replace TTIRED_2_R=0 if TIRED_1==1
tabulate TIRED_2_R

```

**TIRED\_2\_R: How long most recent tired or exhausted feelings lasted?**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	5619	32.4	33.8	33.8
	Some of the day	8036	46.4	48.3	82.0
	Most of the day	1955	11.3	11.7	93.8
	All of the day	1036	6.0	6.2	100.0
	Total	16646	96.1	100.0	
Missing		680	3.9		
Total		17326	100.0		

Step 38. The syntax below recodes TIRED\_3 into TIRED\_3Y to place “SOMEWHERE BETWEEN” numerically in-between “A LITTLE” and “A LOT”.

Also, if response to TIRED\_1 is 1: Never, then TIRED\_3 (Level of tiredness) is not asked. This variable is recoded so these individuals are included in the syntax below.

```

gen TIRED_3Y=1 if TIRED_3==1
replace TIRED_3Y=3 if TIRED_3==2
replace TIRED_3Y=2 if TIRED_3==3
replace TIRED_3Y=. if inlist(TIRED_3, 7, 8, 9)
replace TIRED_3Y=0 if TIRED_1==1
tabulate TIRED_3Y

```

**TIRED\_3Y: Level of tiredness**

		Frequency	Percent	Valid Percent	Cumulative Percent
Valid	Not asked	5619	32.4	33.8	33.8
	A little	4912	28.4	29.5	63.3
	In between a little and a lot	4030	23.3	24.2	87.5
	A lot	2087	12.0	12.5	100.0
	Total	16648	96.1	100.0	
Missing		678	3.9		
Total		17326	100.0		

Step 39. Generate a cross-tabulation of the FATIGUE Extended Set questions: TIRED\_1\_R, TIRED\_2\_R and TIRED\_3Y.

The syntax below produces a cross-tabulation of TIRED\_1\_R: *How often you felt tired or exhausted in the past 3 months* (a measure of frequency) and TIRED\_2\_R: *How long those feelings lasted* (a measure of duration) and TIRED\_3Y: *The level of tiredness* (a measure of intensity) – used to determine a single TIRED INDICATOR (T\_INDICATOR).

`table TIRED_2_R TIRED_1_R , by(TIRED_3Y ) contents(freq)`

TIRED_3Y: Level of tiredness: Intensity	TIRED_2: How long feelings lasted: Duration	TIRED_1: How often felt very tired or exhausted in past 3 months: Frequency				Total
		Never	Some days	Most days	Every day	
Not asked	Not asked	5619				5619
A little	Some of the day		4066	264	124	4454
	Most of the day		252	73	27	352
	All of the day		68	15	18	101
In between	Some of the day		2224	400	166	2791
	Most of the day		497	266	123	887
	All of the day		194	71	84	349
A lot	Some of the day		536	165	84	785
	Most of the day		297	255	160	713
	All of the day		237	118	230	585
<b>TOTAL</b>		5619	8371	1627	1016	16633

Step 40. Create a FATIGUE INDICATOR (T\_INDICATOR) based on the three FATIGUE questions TIRED\_1\_R, TIRED\_2\_R and TIRED\_3Y.

Syntax below creates T\_INDICATOR based on the distribution in the cross-tabulation above.

```
gen T_INDICATOR = .
replace T_INDICATOR=1 if (TIRED_1_R==1)
replace T_INDICATOR=1 if (TIRED_1_R==2 & TIRED_2_R==1 & TIRED_3Y==1)
replace T_INDICATOR=1 if (TIRED_1_R==3 & TIRED_2_R==1 & TIRED_3Y==1)

replace T_INDICATOR=2 if (inlist(TIRED_1_R, 2,3,4) & inlist(TIRED_2_R, 2,3) & ///
TIRED_3Y==1)
replace T_INDICATOR=2 if (inlist(TIRED_1_R, 2,3,4) & TIRED_2_R==1 & TIRED_3Y==2)
replace T_INDICATOR=2 if (TIRED_1_R==2 & TIRED_2_R==2 & TIRED_3Y==2)
replace T_INDICATOR=2 if (TIRED_1_R==4 & TIRED_2_R==1 & TIRED_3Y==1)
```

replace T\_INDICATOR=3 if (inlist(TIRED\_1\_R, 3,4) & TIRED\_2\_R==2 & TIRED\_3Y==2)  
 replace T\_INDICATOR=3 if (inlist(TIRED\_1\_R, 2,3,4) & TIRED\_2\_R==3 & TIRED\_3Y==2)  
 replace T\_INDICATOR=3 if (inlist(TIRED\_1\_R, 2,3,4) & TIRED\_2\_R==1 & TIRED\_3Y==3)  
 replace T\_INDICATOR=3 if (inlist(TIRED\_1\_R, 2,3) & TIRED\_2\_R==2 & TIRED\_3Y==3)  
 replace T\_INDICATOR=3 if (TIRED\_1\_R==2 & TIRED\_2\_R==3 & TIRED\_3Y==3)

replace T\_INDICATOR=4 if (TIRED\_1\_R==4 & TIRED\_2\_R==2 & TIRED\_3Y==3)  
 replace T\_INDICATOR=4 if (inlist(TIRED\_1\_R, 3,4) & TIRED\_2\_R==3 & TIRED\_3Y==3)  
 tabulate T\_INDICATOR

		T_Indicator		Valid Percent	Cumulative Percent
		Frequency	Percent		
Valid	1.00	9949	57.4	59.8	59.8
	2.00	3864	22.3	23.2	83.0
	3.00	2312	13.3	13.9	96.9
	4.00	508	2.9	3.1	100.0
	Total	16633	96.0	100.0	
Missing		693	4.0		
Total		17326	100.0		

## Creating Disability Status Indicators

Type of Disability Indicator		Number of Questions
<b>SS_1</b>	Short Set (SS)	<b>6</b>
<b>Extended Set</b>		
<b>ES_1</b>	SS + <u>Hearing-indicator</u> , <u>Mobility-indicator</u> , <u>Cognition-indicator</u> , <u>Upper Body-indicator</u> + PFAD (4)*	<b>25</b>
<b>ES_2</b>	SS + <u>Hearing-indicator</u> , <u>Mobility-indicator</u> , <u>Cognition-indicator</u> , <u>Upper Body-indicator</u> + AD (4)†	<b>20</b>
<b>Short Set Enhanced</b>		
<b>ES_3</b>	SS + Upper Body-indicator + AD (4)†	<b>12</b>

\* PFAD (4): Pain, Fatigue, Anxiety and Depression Indicators at level 4

† AD (4): Anxiety and Depression Indicators at level 4

**SS\_1: WG Short Set Disability Indicator based on the 6 short set questions**

The syntax below calculates the WG Short Set Disability Indicator based on the six short set questions **SS\_1** at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL.

```
gen SS_1=2
replace SS_1=. if (missing(Vision) & missing(Hearing) & missing(Mobility) & ///
missing(Communication) & missing(SELF_CARE) & missing(Cognition))
replace SS_1=1 if ((Vision==3 | Vision==4) | (Hearing==3 | Hearing==4) | ///
(Mobility==3 | Mobility== 4) | (Communication==3 | Communication==4) | ///
(SELF_CARE==3 | SELF_CARE==4) | (Cognition==3 | Cognition==4))
tabulate SS_1
```

**SS\_1: WG Short Set Disability Identifier**

		Frequency	Percent	Valid Percent	Weighted Estimate
Valid	WITHOUT DISABILITY	14905	86.0	88.8	90.5
	WITH DISABILITY	1872	10.8	11.2	9.5
	Total	16777	96.8	100.0	
Missing		549	3.2		
Total		17326	100.0		

**ES\_1: SS\_1 + Hearing-indicator, Mobility-indicator, Cognition-indicator + Upper Body-indicator + PFAD (4)**

The syntax below calculates an Extended Set Disability Indicator (**ES\_1**) based on 25 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Hearing-, Mobility-, Cognition- and Upper body-Indicators; and severity level 4 for Pain-, Fatigue-, Anxiety- and Depression-Indicators.

```
gen ES_1=2
replace ES_1=. if ///
(missing(SS_1) & ///
(H_INDICATOR < 1 | H_INDICATOR > 4) & ///
(MOB_INDICATOR < 1 | MOB_INDICATOR > 4) & ///
missing(COM_SS) & ///
missing(SC_SS) & ///
(COG_INDICATOR < 1 | COG_INDICATOR > 4) & ///
(UB_INDICATOR < 1 | UB_INDICATOR > 4) & ///
missing(P_INDICATOR) & ///
(T_INDICATOR < 1 | T_INDICATOR > 4) & ///
(ANX_INDICATOR < 1 | ANX_INDICATOR > 4) & ///
(DEP_INDICATOR < 1 | DEP_INDICATOR > 4) )
replace ES_1=1 if (SS_1==1 | ///
```

```
(H_INDICATOR==3 | H_INDICATOR==4) | ///
(MOB_INDICATOR==3 | MOB_INDICATOR==4) | ///
(COG_INDICATOR==3 | COG_INDICATOR==4) | ///
(UB_INDICATOR==3 | UB_INDICATOR==4) | ///
P_INDICATOR==4 | T_INDICATOR==4 | ///
ANX_INDICATOR==4 | DEP_INDICATOR==4)
```

**tabulate** ES\_1

**ES\_1: WG-ES Disability Indicator based on 11 domains and 25 questions**

		Frequency	Percent	Valid Percent	Weighted Estimate
Valid	WITHOUT DISABILITY	13823	79.8	79.8	82.3
	WITH DISABILITY	3503	20.2	20.2	17.7
	Total	17326	100.0	100.0	

**ES\_2: SS\_1 + Hearing-indicator, Mobility-indicator, Cognition-indicator + Upper Body-indicator + AD (4)**

The syntax below calculates an Extended Set Disability Indicator (**ES\_2**) based on 20 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Hearing-, Mobility-, Cognition- and Upper body-Indicators; and severity level 4 for Anxiety- and Depression-Indicators.

```
gen ES_2=2
replace ES_2=. if (missing(SS_1) & ///
(H_INDICATOR < 1 | H_INDICATOR > 4) & ///
(MOB_INDICATOR < 1 | MOB_INDICATOR > 4) & ///
missing(COM_SS) & missing(SC_SS) & ///
(COG_INDICATOR < 1 | COG_INDICATOR > 4) & ///
(UB_INDICATOR < 1 | UB_INDICATOR > 4) & ///
(ANX_INDICATOR < 1 | ANX_INDICATOR > 4) & ///
(DEP_INDICATOR < 1 | DEP_INDICATOR > 4))
replace ES_2=1 if (SS_1==1 | (H_INDICATOR==3 | H_INDICATOR==4) | ///
(MOB_INDICATOR==3 | MOB_INDICATOR==4) | ///
(COG_INDICATOR==3 | COG_INDICATOR==4) | ///
(UB_INDICATOR==3 | UB_INDICATOR==4) | ///
ANX_INDICATOR==4 | DEP_INDICATOR==4)
tabulate ES_2
```

**ES\_2: WG-ES Disability Indicator based on 9 domains and 20 questions**

		Frequency	Percent	Valid Percent	Weighted Estimate
Valid	WITHOUT DISABILITY	14222	82.1	82.1	84.6
	WITH DISABILITY	3104	17.9	17.9	15.4
	Total	17326	100.0	100.0	

**ES\_3: SS\_1 + Upper Body-indicator + AD (4)**

The syntax below calculates the WG Short Set ENHANCED Disability Indicator (**ES\_3**) based on the 12 questions at the recommended cut-off. The level of inclusion is: at least 1 domain/question is coded A LOT OF DIFFICULTY or CANNOT DO AT ALL for the six short set question; severity levels 3 or 4 for the Upper body-Indicators; and severity level 4 for Anxiety- and Depression-Indicators.

```

gen ES_3=2
replace es_3=. if (missing(Vision) & ///
missing(Vearing) & ///
missing(Mobility) & ///
missing(Communication) & ///
missing(SELF_CARE) & ///
missing(Cognition) & ///
missing(UB_INDICATOR) & ///
missing(ANX_INDICATOR) & ///
missing(DEP_INDICATOR) )
replace es_3=1 if ((Vision==3 | Vision==4) | ///
(Hearing==3 | Hearing==4) | ///
(Mobility==3 | Mobility==4) | ///
(Communication==3 | Communication==4) | ///
(SELF_CARE==3 | SELF_CARE==4) | ///
(Cognition==3 | Cognition==4) | ///
(UB_INDICATOR==3 | UB_INDICATOR==4) | ///
ANX_INDICATOR==4 | DEP_INDICATOR==4)
tabulate ES_3
    
```

**ES\_3: WG-SS Enhanced Disability Indicator based on 9 domains and 12 questions**

		Frequency	Percent	Valid Percent	Weighted Estimate
Valid	WITHOUT DISABILITY	14393	83.1	85.8	87.7
	WITH DISABILITY	2384	13.8	14.2	12.3
	Total	16777	96.9	100.0	
Missing		549	3.2		
Total		17326	100.1		



## 补充说明：为什么将疼痛和疲劳排除在外？

值得注意的是，在上述几个残疾标识符中排除了“疼痛”和“疲劳”这两个领域。华盛顿小组内部对这些领域已经进行了相当多的讨论。严格地说，它们不是功能领域（我们的分析表明它们都与其他领域高度相关），并且在纳入这些领域后，残疾发生率可能会非常高。最后，在国际可比性方面，这些领域的普适性较低；也就是说，它们比其他功能领域更容易受到当地社会文化的影响。

出于这些原因，尽管它们可以被纳入国家统计局在全国范围内进行的补充分析中，但我们仍然选择将它们从几个分析中排除。